SysCmdFunktionen optimal nutzen Teil 1
Versionen: Access 2007, 2003, 2002/XP, 2000 und 97

"SysCmd()" fasst gleich mehrere vielseitige und nützliche Funktionen zusammen, die Sie für die unterschiedlichsten Aufgabenstellungen und Problemlösungen einsetzen können. In unserem Tipp zeigen wir Ihnen, was dieses Multitalent im Einzelnen zu bieten hat und wie Sie SysCmd() in Ihren Datenbankanwendungen einsetzen können, um beispielsweise eine Fortschrittsanzeige oder statischen Text in der Statusleiste anzuzeigen, wichtige Informationen über Umgebungsparameter und Zusatzdateien von Access abzurufen, oder um den Status beliebiger Datenbankobjekte zu ermitteln. Sie können also mit SysCmd() den Dialog mit den Anwendern Ihrer Anwendungen optimieren, sich die Entwicklung eigener Funktionen sparen oder unbeabsichtigten Datenverlusten vorbeugen.

Beginnen wir mit der Syntax von SysCmd(): Beim Aufruf der Funktion SysCmd() übergeben Sie jeweils eine interne Konstante, die der Funktion mitteilt, was sie machen soll. Um zum Beispiel den Ordner zu bestimmen, in dem Sie Access installiert haben, verwenden Sie die Anweisung

strDir = SysCmd (acSysCmdAccessDir)

Die Konstante "acSysCmdAccessDir" teilt SysCmd() mit, was von ihr erwartet wird, und das heißt in diesem Fall, dass die Funktion das Installationsverzeichnis liefern soll.

Das ist natürlich nur ein einfaches Beispiel; in den meisten Fällen müssen Sie mehr als nur eine einzige Konstante übergeben. Insgesamt sind je nach auszuführender Funktion noch ein oder zwei weitere Parameter anzugeben.

Beginnen wir mit der "Statuszeile". Sie haben zwei Möglichkeiten, um die Statuszeile mit der SysCmd()-Funktion zu beeinflussen: Sie zeigen statischen Text wie zum Beispiel "Moment bitte, Daten werden importiert..." in der Statuszeile an. Dabei müssen Sie dafür sorgen, dass die Meldung nach Abschluss Ihrer Routine wieder entfernt und die Statuszeile für Access-interne Meldungen freigegeben wird. Die zweite Möglichkeit besteht darin, die Statuszeile mit Hilfe von SysCmd() als Fortschrittsanzeige zu nutzen, um dem Anwender den aktuellen Status eines zeitintensiven Vorgangs anzuzeigen. In diesem Fall muss Ihr Programmcode die Anzeige initialisieren, aktualisieren und auch wieder entfernen.

Für die Ausgabe von statischem Text rufen Sie SysCmd() einmal auf, um den Text anzuzeigen und ein zweites Mal, um den Text wieder zu entfernen. Anfangs übergeben Sie dabei die Konstante "acSysCmdSetStatus" gemeinsam mit dem Text, den Sie anzeigen möchten. Wenn Sie den Text wieder löschen wollen, übergeben Sie der Funktion die Konstante "acSysCmdClearStatus". Das folgende Beispiel zeigt statischen Text an und entfernt ihn wieder, wenn Sie die Meldungsbox bestätigen:

Sub SysCmdTest()

   SysCmd acSysCmdSetStatus, _
          "Text in Statuszeile!"
   MsgBox "Dies ist ein SysCmd()-Test..."
   SysCmd acSysCmdClearStatus

End Sub

Neben der Anzeige von Text in der Statusleiste kann die Funktion SysCmd() während des Ablaufs zeitaufwendiger Vorgänge eine Fortschrittsanzeige in der Statuszeile einblenden. Dazu sind drei Schritte erforderlich:

1. Initialisieren: Um die Funktion für die Fortschrittsanzeige zu initialisieren, übergeben Sie die Konstante "acSysCmdInitMeter", den Text, der links neben der Fortschrittsanzeige erscheinen soll, und den Maximalwert, bei dessen Erreichen die Anzeige 100% entsprechen soll. Alle Aktualisierungen der Fortschrittsanzeige werden dann in Relation zu diesem Maximalwert gesetzt, um den aktuellen Status zu ermitteln und anzuzeigen.

2. Aktualisieren: Nachdem die Funktion für die Fortschrittsanzeige initialisiert wurde, nehmen Sie eine Aktualisierung über die Konstante "acSysCmdUpdateMeter" vor. Wenn Sie beispielsweise bei der Initialisierung den Maximalwert 3000 festgelegt haben, muss der Aktualisierungswert zwischen 0 und 3000 liegen. Bei dem Wert 1500 würde SysCmd dann entsprechend einen Fortschritt von 50% anzeigen.

3. Entfernen: Nachdem Access die Aufgabe, deren Verlauf Sie anzeigen, beendet hat, wird die Fortschrittsanzeige entfernt, indem Sie SysCmd() die Konstante "acSysCmdRemoveMeter" übergeben. Ohne diesen Schritt würde Access die Fortschrittsanzeige weiterhin mit "100% erledigt" anzeigen, obwohl die Aufgabe längst erledigt ist.

Alle drei Aufgaben lassen sich mit einer einzigen Prozedur "Hinweis" abdecken, die Sie zunächst in ein vorhandenes oder neues Modul kopieren:

Sub Hinweis(strMsg As String, Anz As Long)
  Dim X As Variant

  On Error Resume Next
  If Left$(strMsg, 1) = "-" Then 'Zurücksetzen
      X = SysCmd(acSysCmdRemoveMeter)
  ElseIf Left$(strMsg, 1) = "+" Then 'Initialisieren
      X = SysCmd(acSysCmdInitMeter, Mid$(strMsg, 2), Anz)
  Else 'Aktualisieren
      X = SysCmd(acSysCmdUpdateMeter, Anz)
  End If
  DoEvents

End Sub

Am Anfang einer Routine rufen Sie diese Prozedur beispielsweise wie folgt für die Initialisierung auf:

.....
rs.MoveLast
RecCnt = rs.RecordCount
rs..MoveFirst
Hinweis "+Moment bitte, Daten werden importiert...", _
RecCnt
.....

Hier wird ermittelt, wie viele Datensätze ein Recordset umfasst und dieser Wert als Maximalwert mit dem Text "Moment bitte, Daten werden importiert..." an die Prozedur übergeben. Innerhalb der Schleife erfolgt der Aufruf dann zum Beispiel so:


For I = 1 To RecCnt
  Hinweis "", I
  .....
Next I

Hier wird nur der aktuelle Schleifenwert gemäß Zählervariable "I" an die Prozedur übergeben. Nach Durchlauf der Schleife wird die Statuszeile wie folgt zurückgesetzt:

Hinweis "-", 0

Neben den Funktionen zur Steuerung der Statusleiste kann SysCmd() diverse Informationen über die Access-Umgebung liefern. Am effektivsten ist es, wenn Sie anstelle der etwas unhandlichen SysCmd-Aufrufe mit ihren schwierig zu merkenden Konstanten fertige Funktionen verwenden Ein Beispiel für eine solche Hilfsfunktion ist "AccessDir()". Sie ermittelt den Pfad des Ordners, in dem Access (MSACCESS.EXE) installiert ist. Dabei endet der Pfadname immer mit einem Backslash "\":

Function AccessDir()

  AccessDir = SysCmd(acSysCmdAccessDir)

End Function

Die Funktion SysCmd() gibt Ihnen auch die Möglichkeit, die verwendete Access-Version zu ermitteln: Der entsprechende Aufruf verwendet die Konstante "acSysCmdAccessVer" und liefert folgende Ergebnisse

"12.0" für Access 2007
"11.0" für Access 2003
"10.0" für Access 2002/XP
"9.0" für Access 2000 
"8.0" für Access 97
"7.0" für Access 95
"2.0" für Version 2.0.

In eine Funktion gekleidet, die ein numerisches, leichter abfragbares Ergebnis liefert, sieht das dann so aus:

Function AccessVersion() As Integer

  AccessVersion = Val(Syscmd(acSysCmdAccessVer))

End Function 

Die Funktion konvertiert die Rückmeldung von SysCmd() per "Val()" in einen numerischen Wert, den Sie dann zum Beispiel per "If AccessVersion() < 10 Then..." abfragen können.

Weitere nützliche SysCmd()-Funktionen, mit denen Sie beispielsweise Objekte überwachen können, stellen wir in der nächsten Ausgabe vor.

Im ersten Teil dieses ging es um die wichtigsten Grundlagen zur Funktion "SysCmd()" und um einige interessante Funktionen, die wir nun in diesem zweiten Teil weiterführen möchten. So können Sie mit Hilfe von SysCmd() zum Beispiel auch ermitteln, ob die Datenbank mit einer Runtime-Version von Access eingesetzt wird. Dabei hilft Ihnen die Funktion "Runtime()":

Function Runtime()

  Runtime = SysCmd(acSysCmdRuntime)

End Function

Die Funktion liefert als Ergebnis entweder "True/Wahr" oder "False/Falsch". Falls Sie lieber einen Text als Funktionsergebnis zurückliefern möchten, können Sie die folgende Variante der Funktion einsetzen:

Function Runtime()
  Dim varRuntime As Variant

  varRuntime = SysCmd(acSysCmdRuntime)
  If varRuntime = 0 Then
    varRuntime = "Keine Runtime-Version"
  Else
    varRuntime = "Runtime-Version"
  End If
  Runtime = varRuntime

End Function

Namen und Pfad der aktuellen Arbeitsgruppendatei liefert eine weitere Funktion, die ebenfalls auf SysCmd() zurückgreift:

Function AccessWorkGroup()

  AccessWorkGroup = SysCmd(acSysCmdGetWorkGroupFile)

End Function

Ein drittes Einsatzgebiet der SysCmd()-Funktion betrifft die Abfrage von Statusinformationen zu beliebigen Datenbankobjekten. Unter anderem können Sie prüfen, ob ein Objekt neu, geöffnet, geändert oder gespeichert ist. Um beispielsweise festzustellen, ob ein Objekt in Access geöffnet ist, verwenden Sie die Anweisung

intStatus = SysCmd(acSysCmdGetObjectState, _
                   intObjType, _
                   strObjName)

Die Anweisung soll einmal an einem praktischen Beispiel verdeutlicht werden: Wenn Sie beispielsweise den Status eines Moduls mit dem Namen "modInfo" abfragen wollen, müsste der Funktionsaufruf folgendermaßen lauten:

intStatus = SysCmd(acSysCmdGetObjectState, _
                   acModule, _
                   "modInfo")

Es ist nicht immer leicht, zu erkennen, welche Bedeutung das Ergebnis hat. Falls es sich dabei um den Wert "0" (Null) handelt, ist das Objekt nicht geöffnet. Wenn jedoch ein anderer Wert als Null vorliegt, kann der zurückgelieferte Wert die Summe beliebiger oder sogar aller Konstanten sein:

Geöffnet: acObjStateOpen
Neu: acObjStateNew
Geändert, aber nicht gespeichert: acObjStateDirty

Falls ein Objekt neu ist, besteht der zurückgelieferte Wert beispielsweise aus der Summe von "acObjStateOpen" und "acObjStateNew". Sie müssen daher den Operator "AND" verwenden, um die Elemente zu ermitteln, aus denen sich das Ergebnis zusammensetzt. Um festzustellen, ob das Objekt "frmTest" neu ist, verwenden Sie die folgenden Anweisungen:

Dim intStatus As Integer
...
intStatus = SysCmd(acSysCmdGetObjectState, _
                   acForm, _
                   "frmTest")
If intStatus AND acObjStateNew Then
  'Neues Objekt
End If
...

Sie können den Status nicht mit "acObjStateNew" vergleichen, da der Status in Wirklichkeit die Summe aus "acObjStateOpen", "acObjStateNew" und "acObjStateDirty" ist. Wenn Sie den von der Funktion SysCmd() zurückgelieferten Wert mit der Konstanten "acSysCmdGetObjectState" kontrollieren, liefert Access den Wert Null, falls das Objekt nicht geöffnet ist, und in allen anderen Fällen einen Wert größer als Null.

Mit Hilfe des Codes aus unserem nachfolgenden Listing können Sie beispielsweise feststellen, ob ein Formular geöffnet ist. Die Funktion "IsFormOpen()" liefert entweder den Wert "True/Wahr" oder "False/Falsch" zurück:

Function IsFormOpen(strName As String) As Boolean

 IsFormOpen = (SysCmd(acSysCmdGetObjectState, _
                      acForm, strName) <> 0)
End Function

Was aber, wenn Sie Informationen über ein Objekt abfragen möchten, bei dem es sich nicht um ein Formular handelt? In diesem Fall übergeben Sie ganz einfach den Objekttyp (acForm, acTable, acQuery etc.) als optionalen Parameter und können so die Abfrage in einer einzigen Funktion zusammenfassen:

Function IsOpen(strName As String, _
                Optional varObjType As Variant) _
                As Boolean

   If IsMissing(varObjType) Then
      varObjType = acForm
   End If

   IsOpen = (SysCmd(acSysCmdGetObjectState, _
                    varObjType, strName) <> 0)
End Function

Falls beim Aufruf der Funktion kein Objekttyp festgelegt wird, geht die Funktion vom Objekt "acForm" aus.

In vielen Datenbanken muss eine VBA-Routine solange unterbrochen werden oder eine Pause machen, bis ein zuvor geöffnetes Formular oder ein in der Seitenansicht angezeigter Bericht wieder geschlossen ist. Mit der Anweisung

DoCmd.OpenForm "Neuer Datensatz", , , , ,acDialog

lässt sich hier nur bedingt arbeiten, da sie nicht mit Berichten angewendet werden kann - dort gibt es keinen "acDialog"-Modus - und nicht immer alle Formulare als Dialoge angezeigt werden können, sondern auch der Einsatz mehrerer Formulare nebeneinander denkbar ist.

Sie können hier zur Lösung des Problems auf das oben erläuterte Feature "GetObjectState()" von "SysCmd()" zurückgreifen: Über diese Sub-Funktion von SysCmd() lässt sich recht einfach feststellen, ob ein Formular oder ein Bericht geöffnet ist und solange in einer entsprechenden Schleife warten.

Sub WaitForm(strForm As String)

  While SysCmd(acSysCmdGetObjectState, _
               acForm, strForm)
    DoEvents
  Wend

End Sub

Sub WaitReport(strReport As String)

  While SysCmd(acSysCmdGetObjectState, _
               acReport, strReport)
    DoEvents
  Wend

End Sub

Die Funktion SysCmd() prüft hier jeweils über die Sub-Funktion "GetObjectState", ob das Objekt, dessen Name als Parameter übergeben wurde, geladen ist. Wenn ja, ist das Funktionsergebnis "True" und die Schleife wird weiter ausgeführt. Andernfalls ist das Objekt geschlossen worden und die Schleife ist beendet. Es geht dann in der VBA-Routine weiter, aus der Sie die Prozedur aufgerufen haben. Die Anweisung "DoEvents" sorgt jeweils dafür, dass Access innerhalb der Schleife Prozessor-Zeit ans System zurückgibt und so die Performance nicht bremst.

Diese beiden Prozeduren können Sie in Zukunft in VBA-Routinen einsetzen, um mit der Ausführung zu warten, bis ein Formular oder Bericht wieder geschlossen ist:

Private Sub btnTestForm_Click()

  DoCmd.OpenForm "Kunden"
  WaitForm "Kunden"
  MsgBox "Wieder da!"

End Sub

Hier wird zunächst ein Formular geöffnet und anschließend die Prozedur "WaitForm" mit dem Namen des betreffenden Formulars als Parameter aufgerufen. "WaitForm" wartet nun solange, bis das Formular geschlossen ist, anschließend geht es in der VBA-Routine weiter. Sie zeigt hier zu Testzwecken eine kleine Meldung an. Genauso einfach warten Sie auch darauf, dass ein Bericht wieder geschlossen wird:

Private Sub btnTestReport_Click()

  DoCmd.OpenReport "Katalog", acPreview
  WaitReport "Katalog"
  MsgBox "Wieder da!"

End Sub

